




MICROSOFT ACCESS CALENDAR REPORTS

WEEK-AT-A-GLANCE MICROSOFT ACCESS REPORT

OVERVIEW


- Data by itself is of little value. It isn't until we effectively organize, aggregate, and display that data gains meaning
- The quality of visual presentation of data/records helps with our effective and rapid interpretation of complex information
- Colors, fonts, shapes, positions, and other formatting draw our eyes to what's important
- Familiar patterns assist with understanding
- Thorough yet simple

Charles Joseph Minard's Visualization Of Napoleon's 1812 March



SPECIFICATION

- Create week-at-a-glance calendar similar to Outlook
- Vertical position and height dependent on starting time and duration
- Simple with minimal object maintenance
- Data driven flexibility




OTHER SOLUTIONS

- Crystal has a great Calendar Maker
 - Great flexibility (1,000+ lines of code)
 - Most code in standard modules
- Richard Rost and many others have some great stuff but I couldn't find the method I eventually used.
- Power BI has great potential and has made huge breakthroughs in data presentation

Microsoft Access Calendar Seminar

Use Access to Create Monthly Calendars

If you've ever wanted to print monthly **calendars** from your Access databases, then this seminar is perfect for you.



This seminar covers creating an **appointment** database in Microsoft Access. You will create a form that looks like an actual monthly calendar, that you can review your appointments on. Then, when you're ready to print, just pick a month, and the report is generated. Click here for a video showing what's covered in this seminar:

WEEKLY SCHEDULE BY DOCTOR DESIGN

- **Demo** Weekly Schedule by Doctor
- Detail section has controls positioned almost randomly
- No filters have been used in this demo
- Simple black and white
- Doctor then Week of groupings
- Detail and WeekOf footer both 6"

The screenshot shows a report design tool interface for a report named 'rptAppointWeekly'. The design is organized into sections: Page Header, Doctor Header, Detail, and WeekOf Footer. The Detail section contains fields for SchedDate, StartTime, EndTime, Doctor, Patient, and WeekOf. The WeekOf Footer contains formulas like =[WeekOf]+1 and =[WeekOf]+2. A red box highlights the footer formulas with a note: 'These values could be created using the Print method of the report based on data.'

Page Header	
Doctors Smith, Anderson	
Doctor Header	
Detail	
SchedDate	
StartTime	
EndTime	
Doctor	
Patient	
WeekOf	
WeekOf Footer	
= [WeekOf] + 1	= [WeekOf] + 2

WEEKLY SCHEDULE BY DOCTOR CODE

```
Private Sub Detail_Format(Cancel As Integer, FormatCount As Integer)
    Dim lngTopMargin As Long
    Dim lngOneMinute As Long 'size of one minute in twips
    Dim datSchedStart As Date

    datSchedStart = #8:00:00 AM#
    lngOneMinute = 12 'number of twips in one minute
    lngTopMargin = Me.boxTimeLine.Top ' 720 'timeline starts 1/2" down in section
    'create detail sections as 'layers'
    Me.MoveLayout = False
    Me.Patient.Top = lngTopMargin + DateDiff("n", datSchedStart, Me.StartTime) * lngOneMinute
    Me.Patient.Height = DateDiff("n", Me.StartTime, Me.EndTime) * lngOneMinute
    Me.Patient.Left = DateDiff("d", Me.WeekOf, Me.SchedDate) * 2160
End Sub
```

WeekOf Footer						
		=[WeekOf]+1	=[WeekOf]+2	=[WeekOf]+3	=[WeekOf]+4	=[WeekOf]+5
8:00						
9:00						
10:00						
11:00						

These values could be created using the Print method of the report based on data.

rptAppointmentWeekly			
		1	2
Page Header			
Doctors Smith, Anderson			
Doctor Header			
Detail			
SchedDate			
StartTime			
EndTime			
Doctor			
Patient			
WeekOf			

LIMITATIONS & OPPORTUNITIES

- Limitations
 - Isn't designed to handle events outside of standard scheduling
 - Somewhat small text boxes to display names
- Opportunities
 - Dynamically draw time scale
 - Incorporate filtering as needed for date range and doctor
 - Additional information could be added



DAILY SCHEDULE ALL DOCTORS DESIGN

- **Demo** Daily Schedule All Doctors
- Again detail section has controls positioned almost randomly
- ReportColumn is horizontal position of doctor column (twips)
- Simple black and white
- Grouped by SchedDate
- Detail and SchedDate footer both 7"

rptAppointByDoctor

Page Header

= "Doctors " & C

Detail

SchedDate

StartTime

EndTime

Doctor

Patient

ReportColumn

SchedDate Footer

SchedDate


8:00

9:00

10:00

DAILY SCHEDULE ALL DOCTORS CODE

```
Private Sub Detail_Format(Cancel As Integer, FormatCount As Integer)
    Dim lngTopMargin As Long
    Dim lngOneMinute As Long           'size of one minute in twips
    Dim datSchedStart As Date
    datSchedStart = #8:00:00 AM#
    lngOneMinute = 12
    lngTopMargin = 1440                'timeline starts 1" down in section
    Me.MoveLayout = False
    Me.Patient.Top = lngTopMargin + DateDiff("n", datSchedStart, Me.StartTime) * lngOneMinute
    Me.Patient.Height = DateDiff("n", Me.StartTime, Me.EndTime) * lngOneMinute
    Me.Patient.Left = Me.ReportColumn  'field in table
    Me.Doctor.Left = Me.ReportColumn  'field in table
End Sub
```



The screenshot shows a report titled 'rptAppointByDoctor'. The form layout includes a Page Header section with the text '=\"Doctors \" & C'. Below the header is a Detail section with fields for SchedDate, StartTime, EndTime, Doctor, Patient, and ReportColumn. The Doctor and Patient fields are highlighted with black boxes.

LIMITATIONS & OPPORTUNITIES

- Limitations
 - Isn't designed to handle events outside of standard scheduling
 - Somewhat small text boxes to display names
- Opportunities
 - Dynamically draw time scale
 - Incorporate filtering as needed for date range and doctor
 - Additional information could be added similar to task list in Outlook calendar view



CREW ROTATION SCHEDULE DESIGN

- **Demo** Crew Rotation Schedule
- Again detail section has controls positioned almost randomly
- Timeline created across top
- Color used for Vessel
- Grouped by Vessel
- Detail and Vessel footer both 0.3"

[illegible]


CREW ROTATION SCHEDULE CODE

```
Private Sub Detail_Format(Cancel As Integer, FormatCount As Integer)
    Dim lngDuration As Long 'days of tour
    Dim lngStart As Long 'start date of tour
    Dim lngLMarg As Long
    Dim dblFactor As Double
    'put a line control in your page header that starts 1/1 and goes to 12/31
    lngLMarg = Me.boxTimeLine.Left
    dblFactor = Me.boxTimeLine.Width / 365
    lngStart = DateDiff("d", #1/1/2001#, Me.[Start Date])
    lngDuration = DateDiff("d", Me.[Start Date], Me.[End Date])
    'set the color of the bar based on a data value
    Me.txtName.BackColor = Me.VesselColor
    Me.txtName.Width = 10 'avoid the positioning error
    Me.txtName.Left = (lngStart * dblFactor) + lngLMarg
    Me.txtName.Width = (lngDuration * dblFactor)
    Me.MoveLayout = False
End Sub
```

[illegible]


DUAL VALUE CHART DESIGN

- **Demo** Dual Value Chart
- Again detail section has controls positioned almost randomly
- Math is used to calculate height and width of chart bars




DUAL VALUE CHART CODE

```
Private Sub Detail_Format(Cancel As Integer, FormatCount As Integer)
    Dim intSpaceBetween As Integer 'spacing between bars
    Dim intBarHeight As Integer
    Dim intSpacePerEmployee As Integer 'used to create bar widths
    'calculate the bar height based on the percent of maximum and height of box
    intBarHeight = Me.boxBorder.Height * 0.9 * (Me.SalaryGroup / Me.txtMaxSalaryGroup)
    intSpaceBetween = 100
    'calculate the space of a single employee based on width of box, _
    number of employees and spaces between
    intSpacePerEmployee = (Me.boxBorder.Width - (intSpaceBetween * _
    (Me.txtCountGroups + 2))) / Me.txtCountEmps
    'code that overlays all detail sections
    Me.MoveLayout = False
    'find the top of the salary bar
    Me.CountOfSalary.Top = Me.boxBorder.Top + Me.boxBorder.Height - intBarHeight
    'set the height of the bar
    Me.CountOfSalary.Height = intBarHeight
    'set the vertical position of the labels under the bars
    Me.SalaryGroup.Top = Me.boxBorder.Top + Me.boxBorder.Height + 60
    'set the width of the bar
    'set the width prior to the left to avoid error of can't place the control
    Me.CountOfSalary.Width = Me.CountOfSalary * intSpacePerEmployee
    'set the left of the bar
    Me.CountOfSalary.Left = Me.boxBorder.Left + intSpaceBetween * Me.txtCount + _
    (Me.txtCountOfSalaryRunSum - Me.CountOfSalary) * intSpacePerEmployee
    'set the left of the group labels under bars
    Me.SalaryGroup.Left = Me.CountOfSalary.Left
    Me.Total_Salary.Top = Me.CountOfSalary.Top + Me.CountOfSalary.Height / 2.5
    Me.Total_Salary.Left = Me.CountOfSalary.Left + Me.CountOfSalary.Width / 2 _
    - intSpaceBetween * 2
End Sub
```



CALENDAR STYLE SCHEDULE DESIGN

- **Demo** Calendar Style Schedule
- Groups by Doctor and WeekOf
- Uses five copies of the same subreport of a single day
- Five text boxes across top of detail calculate dates which are used as Link Master for the daily subreports
- Uses Line method in **On Print** event of detail to border subreports based on tallest date



CALENDAR STYLE SCHEDULE CODE

```
Private Sub Detail_Print(Cancel As Integer, PrintCount As Integer)
    Dim lngHeight As Long 'store the height of tallest control
    Dim i As Integer
    Dim lngLeft As Long
    For i = 1 To 5
        'compare heights
        If Me("srpt" & i).Height > lngHeight Then
            lngHeight = Me("srpt" & i).Height
        End If
    Next
    'add the height of other control
    lngHeight = lngHeight + Me.txtDay1.Height
    'draw the lines
    For i = 0 To 5
        lngLeft = i * Me.srpt1.Width
        Me.Line (lngLeft, 0)-(lngLeft, lngHeight)
    Next
End Sub
```

LIMITATIONS & OPPORTUNITIES

- Limitations
 - Much of the design is hard-code for simplicity sake
 - It's on paper and not dynamic
 - Only works where code runs (Print Preview and Run Report)
 - Doesn't support odd shapes without a lot of work
- Opportunities
 - Might be used for seating charts, cemetery plots, org charts, graphs, etc.
 - Make more dynamic and use the report's Print method with CurrentX and CurrentY

Chelsea [P] Smith	Diana [3] Smith	Stephanie [3] Smith	1	Edwin [4] Smith	Axel [1] Smith
Paula* [K] Smith	Kassandra [3] Smith	Julio [4] Smith	2	Jacob [2] Smith	Omar [1] Smith
Gabriel [P] Smith	Valeria [5] Smith	Joshua [3] Smith	3	Jesus [4] Smith	Antoni [1] Smith
Emily [K] Smith	Jaqueline [2] Smith	Lizeth [K] Smith	4	Brisseyda [3] Smith	Barbara [4] Smith
	Manuel [5] Smith	Melanei [3] Smith	5	Tracey [1] Smith	Natalia [3] Smith
Dan [4] Smith	Jesus [5] Smith	Neli [3] Smith	6	Ben* [4] Smith	Ruby [2] Smith
Johnathan [5] Smith		Heriberto [3] Smith	7	107	106
Ashley [1] Smith	David [2] Smith	David [2] Smith	8	96	95
Samantha [5] Smith		Amy [5] Smith	9	85	84
Stephanie [5] Smith		Angela [5] Smith	10	78	77
Alex [4] Smith		Angel [2] Smith	11	69	68
Jennifer [3] Smith	Ignacio [K] Smith	Rodolfo [4] Smith	12	58	57
			13	49	48
				39	38
				29	28
				19	18
				9	8
				1	2

QUESTIONS - DISCUSSION

